
Overview 
-- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

 
正しくない引数リストを渡してカスタムメソッドをコールすると、エラーが表示されたり、

クラッシュしたり、データの扱いが不正になったりします。幸いなことにこれらの問題はシ

ンプルな防御コードを使用して避けることが可能です。このテクニカルノートでは、不正な

数の引数を防がなかった場合の4th Dimensionの動作を確認し、次に不正な引数リストを検

知、管理する汎用的なメソッドを提供します。 
 
 
Example Method and 4th Dimension Behavior 
-- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

 
不正な引数リストに起因する問題を議論する前に、存在しない引数を参照したときの4th 
Dimensionの振る舞いについて見ていきましょう。この議論では、以下のメソッド

DisplaySumを例とします: 
 

C_LONGINT($1;$firstValue_l) 
C_LONGINT($2;$secondValue_l) 
 
$firstValue_l:=$1 
$secondValue_l:=$2 
 
C_LONGINT($sum_l) 
$sum_l:=$firstValue_l+$secondValue_l 
 
ALERT(String($sum_l)) 

 
コードは引数として渡された二つの倍長整数を加算し、結果を表示します。例題コードは意

図的に簡単にしてあります。以下はこのメソッドを正しい引数リストでコールするサンプル

です: 
 
DisplaySum (1;2) 
 
DisplaySumメソッドは$1と$2の値を加算し、以下のような結果を表示します: 
 

 
 
必要な引数を渡さなかった場合はどうなるでしょうか? 例えば、以下のようにひとつしか引

数を渡さないと、4th Dimensionはどうするでしょう: 
 
DisplaySum (1) ` 二つの引数が必要なメソッドに一つだけ渡す 
 



この場合DisplaySumの以下のコードが不正となります: 
 
$secondValue_l:=$2 
 
DisplaySumのこの行を実行する際に何が起こるかは、オペレーティングシステムや4th 
Dimensionのバージョン、そしてコードをコンパイルしたときのオプションによって異なり

ます。 
 
これより、さまざまなケースで4th Dimensionがどのような反応を示すか見ていきます. 
 
 
Bad Method Calls: Interpreted 
 
インタープリタモードで動作している場合、問題のコードを実行すると以下のようなシンタ

ックスエラーダイアログが表示されます: 
 

 
 
表示されているエラーは、$2に値が渡されなかったことに対する状況を表現しています。首

尾一貫して、システムの中の関連するすべてのコードでこの種のエラーを一掃するようにコ

ードを記述することができます。しかし効果のあるテストコードをプロシージャに配置して

も、不正な引数リストでメソッドをコールするように書くことは、比較的簡単です。 
 
 
Bad Method Calls: Compiled without Range Checking on Windows 
 
不正な引数リストが引き起こす結果は、インタプリタよりもコンパイルされたデータベース

でより深刻です。コンパイルされたプログラムでの実際の動作はさまざまですが、いずれも

望ましいものとはいえません。例えば以下は、範囲チェックを行わない設定でコンパイルし

たデータベースを、Windows XP上で起動し、問題のコードを実行したときに発生するもの

です: 
 



 
 
このようなダイアログは誰も見たくないものです。何百、何千行のコードの中で、たった一

つの不正な引数リストのために、システムがクラッシュしてしまいます。よってこのような

問題に対処することは価値のあることです。特に上記のクラッシュダイアログのように、問

題の原因についてほとんど何も教えてくれないような問題が発生するケースではなおさらで

す。4th Dimensionにバグがあるのか、プラグインか、あるいは開発者のコードに問題があ

るのか、このダイアログからは知ることができません。例えば、以下の画像はclick here
リンクをクリックしたときに表示される技術情報です: 
 

 
 
残念なことに、上記エラースクリーンには関連する詳細情報や、問題点を特定するような情

報はなにもありません。 
 
 
Bad Method Calls: Compiled without Range Checking on OS X 
 
クラッシュが発生すれば、少なくともシステムに問題が発生していることがわかります。範

囲チェックなしでコンパイルした問題のコードをMac OS Xで実行すると、問題はよりわか

りづらくなります: 
 



 
 
今回のシンプルな例では、1が1584と表示され、誤りがあったことがすぐにわかります。し

かしコードが複雑な計算を行うものであればどうでしょう? データは微妙に間違ったものに

なる可能性があります。これと比べれば、クラッシュが起きたほうが、注意すべき出来事が

発生したのが明瞭であり、ましだといえます。 
 
 
Bad Method Calls: Compiled with Range Checking 
 
範囲チェックが有効になっていれば、4th Dimensionはインタプリタモードと同じようなダ

イアログを表示しようとします: 
 

 
 

 



 
この警告は、クラッシュしたり、誤った結果を表示するよりはより助けとなるものです。OS 
Xでのダイアログがたとえ読めないものであったとしても、プログラムのどこかがおかしいこ

とがわかります。このようなアラーとは、データが壊されてしまう前に、バグを発見する機

械を提供します。 
 
Tip コンパイルは範囲チェックをオンにして行う。実行速度の差は小さなものであり、非常にまれな
ケース以外でそのことを気にする必要はありません。 
 
 
About the Behavior Documented Above 
 
今まで説明した、不正な引数リストに対する4th Dimensionの振る舞いをあてにすることは

できません。将来のバージョンでは動作が変わることもありえます。 
 
今まで見てきた振る舞いの中には、ほとんど妥当と思われるものはありませんでした。よく

て、4th Dimensionは行儀よく実行を失敗することくらいしかできません。 
 
それよりも、不正な引数リストは開発者のバグであり、発見され、修正されるべきものです。 
 
 
Detecting and Stopping Bad Parameter Lists 
-- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

 
Counting Parameters within Each Method 
 
失われた引数の参照や代入を防ぐ方法は、それらを使用する前に引数の数を数えることで簡

単に確認できます: 
 

If (Count parameters=2) 
  $firstValue_l:=$1 
  $secondValue_l:=$2 
 
  C_LONGINT($sum_l) 
  $sum_l:=$firstValue_l+$secondValue_l 
  ALERT(String($sum_l)) 
Else 
  ALERT(“DisplaySum was called with the wrong number of parameters.") 
End if 

 
引数を二つ以外渡してこのメソッドをコールすると、このメソッドは以下のようなアラート

を表示します: 
 



 
 
簡単なIf テストとALERT がこのコードをよりよいものにしています。このコードではクラ

ッシュや不正な結果が生じることはありません。画面には問題の箇所が明示的に表示されま

す。 
 
Tip ALERT コマンドを使用しないことも可能です。このコマンドを使用できない状況ではエラーをロ
グに記録したり、返り値として返したり、現在のコンテキストにあった方法で処理します。このよう
な状況は例えばトリガやWeb、Webサービスでコードが実行されているときに当てはまります。 
 
 
Using a Centralized Parameter Tester 
 
先に示したコードは失われた引数を検知するのに役立ちますが、エラーを処理するためのコ

ードが直接メソッドに記述されています。 
 
多くの開発者にとって、作業を行うコードとエラーを管理するコードを分離させたほうが便

利で自由度が高くなります。例えば以下のメソッドを見てください: 
 

C_LONGINT($1;$firstValue_l) 
C_LONGINT($2;$secondValue_l) 
 
If (ParameterCountIsOkay (Current method name;2;2;Count parameters)) 
  $firstValue_l:=$1 
  $secondValue_l:=$2 
   
  C_LONGINT($sum_l) 
  $sum_l:=$firstValue_l+$secondValue_l 
  ALERT(String($sum_l)) 
End if 

 
直接引数をカウントするのではなく、関数ParameterCountIsOkayが呼ばれています。サブル

ーチンは呼び出しもとのメソッド名、必要とされる引数の最小数、必要とされる引数の最大

数、そして実際に受け取った引数の数を渡されます。ParameterCountIsOkayはこれらの引数

をテストし、引数の数が範囲に収まっているか検査します。もし範囲外にあれば、以下のよ

うなダイアログを表示します: 
 



 
 
ParameterCountIsOkay ルーチンのコードは以下のようになります: 
 

C_BOOLEAN($0;$countIsOkay_b) 
C_TEXT($1;$methodName_text) 
C_LONGINT($2;$min) 
C_LONGINT($3;$max) 
C_LONGINT($4;$count) 
 
$countIsOkay_b:=False` Default. 
If (Count parameters>=4)` Test that this routine has enough parameters! 
  $methodName_text:=$1 
  $min:=$2 
  $max:=$3 
  $count:=$4 
 
  If (($count<$min) | ($count>$max))` Not enough parameters | Too many parameters 
    $countIsOkay_b:=False` This is a good place to log or display an error. 
 
    C_TEXT($error_text) 
    $error_text:="" 
    $error_text:=$error_text+Char(Double quote)+$methodName_text+Char(Double quote) 
    $error_text:=$error_text+" called with "+String($count)+" parameter(s). " 
    $error_text:=$error_text+Char(Carriage return)+Char(Carriage return) 
    $error_text:=$error_text+"At least "+String($min)+ 
    $error_text:=$error_text+" required and at most "+String($max)+" allowed." 
    ALERT($error_text) 
 
  Else 
    $countIsOkay_b:=True 
  End if 
 
Else ` The ParameterCountIsOkay routine didn't get enough parameters. 
    ` This is a good place to log or display an error. 
  C_TEXT($error_text) 
  $error_text:="" 
  $error_text:=$error_text+" called with "+String($count)+" parameter(s). " 
  $error_text:=$error_text+Char(Carriage return ) 
  $error_text:=$error_text+"Four parameters are required" 
  ALERT($error_text) 
End if `(Count parameters>=4) 
 
$0:=$countIsOkay_b 

 
このように関数にロジックを移動することにはいくつかの利点があります: 
 
・個々のメソッドに対して引数を使用する前にIfテストを追加することが簡単になります。そ

れぞれのメソッドが個別にエラーを処理する必要がなくなり、その分シンプルになります。 



 
・コードのテスト、デバッグ、管理がより簡単になります。 
 
・コードを集中させることで、新しい機能の追加が簡単にできます。例えばログ機能の実装やエラー
のメール機能、あるいはトリガや Web/SOAP プロセスでエラーダイアログを出さないようにするなど
をひとつのメソッドの中で処理できます。 
 
 
Regarding ON ERR CALL 
 
ON ERR CALL コマンドを使用して、エラー管理処理に独自のメソッドを挿入することがで

きるようになります。エラーが検知されると、ON ERR CALLでインストールされたメソッ

ドが実行されます。プログラムはクラッシュせず、範囲チェックダイアログやシンタックス

エラーダイアログも表示されません。残念なことに、ON ERR CALLにはドキュメント化さ

れていない制限があり、すべてのエラーをトラップするわけではありません。 
 
例えば、ON ERR CALL は不正な引数リストによるコンパイルモードでのエラー処理につい

てはうまく管理できません。 
 
 
Summary 
-- - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - 

 
カスタムメソッドで渡されない引数を参照するとエラーダイアログが表示されたり、クラッ

シュしたり、結果が誤るなどの現象が発生します。これらの問題は簡単のコードで防ぐこと

が可能です。このテクニカルノートでは、不正な引数リストで引き起こされる問題を確認し、

それを完全に避ける方法について紹介しました。 


